Römische Zahlen ausgeben
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

 FRAGE:  Ich muss in einigen Formularen und Berichten römische Zahlen (Quartale, Jahreszahlen usw.) ausgeben. Bei der Funktion "Format()" finde ich keine Maske, um Zahlen in diese Darstellungsart konvertieren zu können. Habe ich etwas übersehen oder gibt es eine andere Lösung?
L. Marrando

 ANTWORT:  Mit der Funktion "Format()" können Sie mit einem kleinen Trick lediglich Quartalsangaben in römischen Zahlen ausgeben:

intQuartal = Format([Datumsfeld],"q") 'Ergebnis ist 1..4
strQuartal= "Quartal " & _
            Choose(intQuartal,"I", "II", "III", "IV")

Um generell numerische Werte in römischen Zahlen auszugeben, kopieren Sie folgende Funktion in eine neues oder in ein vorhandenes Modul:

Public Function DezToRoemisch(ByVal aValue As Long) _
                As String
  Dim I&
  Dim varArabisch As Variant, varRoemisch As Variant
  Dim strResult As String

  varArabisch = _
     Array(1, 4, 5, 9, 10, 40, 50, 90, 100, _
           400, 500, 900, 1000)
  varRoemisch = _
     Array("I", "IV", "V", "IX", "X", "XL", _
           "L", "XC", "C", "CD", "D", "CM", "M")

  For I = UBound(varArabisch) To _
          LBound(varArabisch) Step -1
    Do While aValue >= varArabisch(I)
      aValue = aValue - varArabisch(I)
      strResult = strResult & varRoemisch(I)
    Loop
  Next I
  DezToRoemisch = strResult

End Function

Diese Funktion rufen Sie dann zum Beispiel in VBA wie folgt auf:

strQuartal = DezToRoemisch(Format([Datumsfeld],"q"))

